

2nd Order Differential Equations

Dowling18A.wmx

TABLE OF CONTENTS

PREFACE	1
REFERENCES	1
FAST TRACK ODE2 METHOD	2
NUMERICAL INITIAL VALUE PROBLEMS	3
GENERAL SOLN, 2ND ORDER ODE'S	8
MARKET MODEL WITH PRICE EXPECTATIONS	24
ODE'S WITH A VARIABLE TERM	33
HIGHER ORDER ODE'S WITH DESOLVE	34

```
(%i5) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2],
    background_color = light_gray, draw_realpart=false)$
    fpprintprec:5$ ratprint:false$ kill(all)$
```

```
(%i1) load ("Econ2.mac");
```

```
(%o1) c:/work5/Econ2.mac
```

1 Preface

This worksheet is one of a number of wxMaxima files available in the section
Economic Analysis with Maxima
on my CSULB webpage.

We use Maxima to solve a few of the problems (and draw some of the plots) in Ch. 18 of the supplemental text: Introduction to Mathematical Economics, 3rd ed, (Schaum's Outline Series), by Edward T. Dowling (1992, 2001), McGraw-Hill. This modestly priced text is a bargain with many worked out examples. You should compare the examples worked out "by hand" in this text with what we do using Maxima. Section numbers [x.y] refer to sections in Dowling's text.

We have slightly changed some of the symbols used in particular problems.

Ted Woollett
<https://web.csulb.edu/~woollett/>
 Oct. 26, 2021

2 References

Fundamental Methods of Mathematical Economics, Fourth Ed., A. C. Chiang, and K. Wainwright, McGraw-Hill, 2005.

3 **Fast Track, ode2(eqn,y,t), check (y, t)**

Fast track linear 2nd order ode with constant coefficients.

Standard form: $y'' + A y' + B y(t) = C$.

For ease of checking solution, write eqn with C brought over to the left hand side with a minus sign:

eqn : 'diff (y,t,2) + A*diff(y,t) + B*y - C;

Our Maxima function in Econ2.mac is:

Lode2(y,t,type,A,B,C), with type = real, complex, or double, depends on A and B:

The characteristic roots are solutions of $r^2 + A*r + B = 0$.

distinct real roots if $A^2 > 4*B$

complex roots if $A^2 < 4*B$

double real root if $A^2 = 4*B$.

(%i2) eqn : 'diff(y, t, 2) + 9*diff (y,t) + 14*y - 7;

(eqn)
$$\frac{d^2}{d t^2} y + 9 \left(\frac{d}{d t} y \right) + 14 y - 7$$

For numerical coefficients A, B, Maxima's ode2 returns a solution without asking questions. If A and/or B are symbolic, ode2 asks questions (see examples later).

(%i3) Msoln : ode2 (eqn,y,t);

(Msoln)
$$y = \%k1 \%e^{-2 t} + \%k2 \%e^{-7 t} + \frac{1}{2}$$

Solution assuming values of y and y' at t = 0:

(%i4) Mpsoln : ic2(Msoln,t=0,y=10,'diff(y,t) = -2);

(Mpsoln)
$$y = \frac{129 \%e^{-2 t}}{10} - \frac{17 \%e^{-7 t}}{5} + \frac{1}{2}$$

Using our Lode2 function (Econ2.mac) you must say what kind of roots you expect.

(%i5) Lsoln : Lsoln : Lode2(y,t,real,9,14,7);

this assumes that 81 > 56

(Lsoln)
$$y = \%k1 \%e^{-2 t} + \%k2 \%e^{-7 t} + \frac{1}{2}$$

```
(%i6) Lpsoln : ic2(Lsoln,t=0,y=10,'diff(y,t) = -2);
```

```
(Lpsoln) 
$$y = \frac{129 \%e^{-2t}}{10} - \frac{17 \%e^{-7t}}{5} + \frac{1}{2}$$

```

A solution returned by the Maxima function `ode2` (we used `ode2` for first order ordinary differential equations in Dowling16.wmx) should never need to be checked for accuracy (since why use Maxima if it is not always reliable), but just to show off Maxima tricks, we will define `check(y,t)` here, which should produce zero (0) if `y` is replaced by an actual solution as a function of `t` (given our way of defining `eqn`).

If you have defined `eqn` with the constant `C` term brought over to the lhs with a minus sign (as above), this function should produce 0.

```
(%i7) check(y, t) := expand (ev (eqn,diff));
```

```
(%o7) check ( y , t ) := expand ( ev ( eqn , diff ) )
```

```
(%i8) ys : rhs (Lpsoln);
```

```
(ys) 
$$\frac{129 \%e^{-2t}}{10} - \frac{17 \%e^{-7t}}{5} + \frac{1}{2}$$

```

```
(%i9) check(ys,t);
```

```
(%o9) 0
```

What if you made the wrong type assumption for a numerical case?

```
(%i10) soln : Lode2(y,t,complex,9,14,7);
```

this assumes that 56 > 81

```
(soln) assumption incorrect
```

4 Numerical Initial Value Problems, `ode2`, `ic2`

4.1 Dowling Prob. 18.8

Given the 2nd order ODE: $y'' + 9y' + 14y = 7$, find the particular (definite) solution for which $y(0) = -5/2$ and $y'(0) = 31$. Comment on stability.

This example illustrates the required order of arguments for the Maxima functions `ode2` & `ic2`. Notice the use of the single quote mark (') in front of `diff`, to prevent immediate action by the Maxima function `diff`, in the definition of `eqn`.

```
(%i13) eqn : 'diff(y, t, 2) + 9*diff (y,t) + 14*y - 7;
        gsoln : ode2 (eqn, y, t);
        psoln : ic2 (gsoln, t = 0, y = -5/2, 'diff (y,t) = 31 );
```

$$(eqn) \quad \frac{d^2}{dt^2} y + 9 \left(\frac{d}{dt} y \right) + 14 y - 7$$

$$(gsoln) \quad y = \%k1 \%e^{-2 t} + \%k2 \%e^{-7 t} + \frac{1}{2}$$

$$(psoln) \quad y = 2 \%e^{-2 t} - 5 \%e^{-7 t} + \frac{1}{2}$$

```
(%i14) grind(%)$
        y = 2*%e^-(2*t)-5*%e^-(7*t)+1/2$
```

```
(%i15) check (rhs (psoln),t);
```

```
(%o15) 0
```

4.1.1 yc, yp

If the ODE has the form $y'' + A y' + B y = C$, the "complimentary solution" y_c is $y_c = A_1 \exp(r_1 t) + A_2 \exp(r_2 t)$, in which r_1, r_2 are the roots of the quadratic equation $r^2 + A r + B = 0$. This equation comes from using for $y(t)$ the trial solution

$y = \exp(r t)$ in the "reduced equation" gotten by setting $C = 0$:

$y'' + A y' + B y = 0$, whence

$(r^2 + A r + B) \exp(r t) = 0$, but $\exp(r t) \neq 0$, so we must have $r^2 + A r + B = 0$ for this to work.

For this example, these roots are:

```
(%i16) solns : solve (r^2 + 9*r + 14, r);
```

```
(solns) [r=-2,r=-7]
```


The general solution (with the two arbitrary adjustable constants $\%k1$ and $\%k2$) has the structure of $y_{gen} = y_c + y_p$, in which y_p is the "particular solution" $y_p = C/B$, which we can call the "equilibrium solution" (hopefully) since if we assume we have waited so long that both the first and second derivatives of $y(t)$ equal zero, the given ODE reduces to $B y = C$, or $y(t) = C/B =$ independent of time. In our example, $C/B = 7/14 = 1/2$.

Since both "characteristic roots" r_1 and r_2 are negative numbers, and $\exp(-|r| t) \rightarrow 0$ as $t \gg 1/|r|$, the general solution will approach the particular solution $y_p = 1/2$ if we wait long enough ($t \gg 1/7 = 0.143$). "much greater" (\gg) means, say, ten or twenty times, so $t > 1.4$ as a rough estimate.

```
(%i17) ys : rhs (psoln);
```

```
(ys) 2 %e-2 t - 5 %e-7 t +  $\frac{1}{2}$ 
```

```
(%i18) wxdraw2d (xlabel = "t", ylabel = "y", key_pos = bottom_right,
  key = "y(0) = -5/2, dy/dt(0) = 31", explicit (ys, t, 0, 2),
  color = red, key = "ye = 1/2", explicit (1/2, t, 0, 2),
  color = black, key = "", line_width = 1, explicit (0, t, 0, 2))$
```



4.1.2 Characteristic Roots Notation: g, h

Maxima returns two solutions of the equation $r^2 + A r + B = 0$:

```
(%i19) solve (r^2 + A*r + B, r);
```

```
(%o19) [r =  $-\frac{\sqrt{A^2 - 4 B} + A}{2}$ , r =  $-\frac{\sqrt{A^2 - 4 B} - A}{2}$ ]
```

We adopt a notation based on Dowling's treatment:

$r_1, r_2 = g \pm i h$, or in Maxima notation, $g \pm \%i h$, where $\%i = \sqrt{-1}$, $(\%i)^2 = -1$, and $g = -A/2$, $h = \sqrt{4*B - A^2}/2$, connected with our chosen notation for a general linear 2nd order ODE: $y'' + A y' + B y(t) = C$.

For this example, $A = 9$, $B = 14$, so $g = -9/2$, and $h = \sqrt{4*B - A^2}/2 = \sqrt{56 - 81}/2 = \%i \sqrt{81 - 56}/2 = \%i (5/2)$, so $r_1 = g + \%i h = -9/2 + (\%i)^2 5/2 = -9/2 - 5/2 = -14/2 = -7$, and $r_2 = g - \%i h = -9/2 + 5/2 = -4/2 = -2$.

Since both r_1 and r_2 are negative (real) numbers, the solution $y(t)$ will approach a stable solution for "large" positive values of t .

4.1.3 croots (A,B)

Let `croots (A,B)` be a Maxima function which returns a list of characteristic roots in the form `[r = r1, r = r2]`.

```
(%i20) croots(%A,%B) :=
      block ([gg,hh],
            gg : -%A/2,
            hh : sqrt (4*%B - %A^2)/2,
            [r = gg + %i*hh, r = gg - %i*hh])$
```

```
(%i21) croots (9,14);
(%o21) [-7,-2]
```

We can get a list of the right hand sides of this list using `map('rhs, alist)`.

```
(%i22) map ('rhs, %);
(%o22) [-7,-2]
```

We can define `r1` and `r2` via a list as follows:

```
(%i23) [r1, r2] : map('rhs, croots(9,14));
(%o23) [-7,-2]
```

Try interrogating this list:

```
(%i24) [r1,r2];
(%o24) [-7,-2]
```

Try interrogating just `r1`:

```
(%i25) r1;
(%o25) -7
```

4.1.4 r1r2 (A, B)

Let's define a second Maxima function that uses this mapping to define `r1` and `r2`.

```
(%i26) r1r2(AA, BB) := map ('rhs, croots (AA, BB))$
```

To check that this function works, we first remove Maxima's knowledge of our prior assignment of values to the symbols `r1` and `r2`, by using the Maxima function `kill`.

```
(%i27) kill(r1,r2)$
```

Now Maxima should not know any value for r1 and r2.

```
(%i28) [r1,r2];
```

```
(%o28) [r1,r2]
```

Now use our new Maxima function r1r2 (A, B) to define r1 and r2:

```
(%i29) [r1,r2] : r1r2 (9, 14);
```

```
(%o29) [-7, -2]
```

```
(%i30) r1;
```

```
(%o30) -7
```

4.2 Dowling Prob 18.9

Find the definite integral of the 2nd order ODE

$$y'' - 4y' - 5y = 35$$

such that $y(0) = 5$ and $y'(0) = 6$.

The characteristic roots r_1, r_2 of the characteristic equation

$r^2 + A r + B = 0$, or $r^2 - 4 r - 5 = 0$, are given by

```
(%i31) solve (r^2 - 4*r - 5, r);
```

```
(%o31) [r=5, r=-1]
```

Or, we can use our Maxima function r1r2 (A, B):

```
(%i32) [r1,r2] : r1r2 (-4, -5);
```

```
(%o32) [-1, 5]
```

Since we have one positive root, that will lead to explosive increase with time, and NOT approach $y_e = 35/(-5) = -7$ for "large" times.

Reverting to the more general approach, we use the Maxima function ode2.

The symbol eqn is defined here using 'diff (...), putting a single quote (') mark in front of the Maxima function diff, which prevents diff from executing immediately.

gsoln is defined to be the general solution with two arbitrary constants %k1 and %k2, and dsoln is the "definite" solution corresponding to the $t = 0$ values of y and y' .

```
(%i35) eqn : 'diff(y, t, 2) - 4*'diff (y,t) - 5*y = 35;
      gsoln : ode2 (eqn, y, t);
      dsoln : ic2 (gsoln, t = 0, y = 5, 'diff (y,t) = 6 );
```

$$(eqn) \quad \frac{d^2}{dt^2} y - 4 \left(\frac{d}{dt} y \right) - 5y = 35$$

$$(gsoln) \quad y = \%k1 \%e^{5t} + \%k2 \%e^{-t} - 7$$

$$(dsoln) \quad y = 3 \%e^{5t} + 9 \%e^{-t} - 7$$


Let ys be a Maxima expression, depending on the time t , which is the desired solution for $y(t)$.

```
(%i36) ys : rhs (dsoln);
```

$$(ys) \quad 3 \%e^{5t} + 9 \%e^{-t} - 7$$

```
(%i37) wxdraw2d (xlabel = "t", ylabel = "y", xrange = [-10, 35],
      key_pos = top_left,
      key = "y(0) = 5, dy/dt(0) = 6", explicit (ys, t, 0, 0.5),
      color = red, key = "ye = -7", explicit (-7, t, 0, 0.5),
      color = black, key = "", line_width = 1, explicit (0,t,0,0.5))$
```

(%t37)



which shows explosive (exponential) growth.

5 $y''(t) + A y'(t) + B y(t) = C$, General Solution from ode2

Consider the "2nd ODE with constant coefficients"

$$d^2y/dt^2 + A dy/dt + B y(t) = C,$$

in which we assume A, B, and C are each real constants.

We define eqn as an expression (which equals zero for later convenience) in a form which ode2 can understand. The single quote mark (') in front of diff prevents diff from going through the actions normally taken by the diff function (technically, the single quote mark "prevents evaluation", producing a "noun form" rather than a "verb form").

(%i38) eqn : 'diff (y,t,2) + A*diff(y,t) + B*y - C;

(eqn)
$$\frac{d^2}{dt^2} y + A \left(\frac{d}{dt} y \right) + B y - C$$

A "general solution" $y(t) = y_c(t) + y_p$.

The "particular solution" y_p is proportional to C and may or may not depend on t.

If $C = 0$, then $y_p = 0$.

If $C \neq 0$ then

1.) if $B \neq 0$, then $y_p = C/B$, which is gotten by assuming y has ceased changing with time in the sense that both the first and second derivative of y wrt t are zero. In that case the given ODE reduces to $B*y = C$.

2.) if $B = 0$, and $A \neq 0$ then $y_p = C*t/A$, which is gotten by assuming $y'' = 0$, in which case the ODE reduces to $dy/dt = C/A$, and the indefinite integral is $y = C*t/A$.

3.) if $B = 0$ and $A = 0$ then $y_p = C*t^2/2$, since the ODE reduces to $y'' = C$, the first indefinite integral is $y' = C*t$, and the indefinite integral of y' is $C*t^2/2$.

The "complementary solution" $y_c = A1*\exp(r1*t) + A2*\exp(r2*t)$, in which $r1$ and $r2$ are the solutions of the quadratic equation $r^2 + A*r + B = 0$, and $A1$ and $A2$ are constants which can be chosen to fit initial conditions (for example).

(%i39) solns : solve (r^2 + A*r + B, r);

(solns)
$$\left[r = -\frac{\sqrt{A^2 - 4B} + A}{2}, r = \frac{\sqrt{A^2 - 4B} - A}{2} \right]$$

CASE $A^2 > 4B$

If $A^2 > 4B$, r_1 and r_2 are real roots and depending on the signs of r_1 and/or r_2 there is a possibility that the complementary solution can grow exponentially (vide infra).

Let $s = \sqrt{A^2 - 4B}$, then $r_1 = -(s + A)/2$, and if $(s + A) < 0$ [which is equivalent to $s < -A$] then $r_1 > 0$ and $\exp(r_1 t)$ grows exponentially. If $(s + A) > 0$ [equivalent to $s > -A$], then $r_1 < 0$ and $\exp(r_1 t)$ decays exponentially to zero.

CASE $4B > A^2$

If, on the other hand, $4B > A^2$, then r_1 and r_2 are complex roots.

Let $p = \sqrt{4B - A^2}$ and use $\sqrt{-1} = i$ and $\sqrt{-x} = i\sqrt{x}$. Then we get $r_1 = -A/2 - \sqrt{-(4B - A^2)}/2 = -A/2 - i\sqrt{4B - A^2}/2 = -A/2 - ip/2$.

Similarly, $r_2 = -A/2 + ip/2$.

Then $y_c = \exp(-A^*t/2)*(A1*\exp(-i*p*t/2) + A2*\exp(i*p*t/2))$.

Use $\exp(i\theta) = \cos(\theta) + i\sin(\theta)$, and $\exp(-i\theta) = \cos(\theta) - i\sin(\theta)$, to replace the complex exponentials with trig functions, replace $A1$ and $A2$ by a different pair of constants $k1$ and $k2$ to get $\exp(-A^*t/2)$ times a linear combination of trig functions with the argument $p^*t/2$.

If $A > 0$, the overall factor $\exp(-A^*t/2)$ drives the whole term to zero for $t \gg 1/A$.

If $A < 0$, the factor $\exp(-A^*t/2)$ shows explosive growth.

If $A = 0$, y_c is an eternally oscillating function of time within fixed limits.

5.1 Case: $A^2 < 4B$, Complex Roots, `Lode2(y,t,type,A,B,C)`

This is the simpler case in which the characteristic roots r_1 and r_2 are complex, and the nature of the complementary solution depends crucially on the sign of A . The Maxima function `ode2` asks us immediately about the sign of $(4B - A^2)$.

(%i40) `gsoln : ode2 (eqn,y,t);`

Is $4B - A^2$ positive, negative or zero? positive;

(gsoln)
$$y = \%e^{-\frac{A t}{2}} \left(\%k1 \sin\left(\frac{\sqrt{4 B - A^2} t}{2}\right) + \%k2 \cos\left(\frac{\sqrt{4 B - A^2} t}{2}\right) \right) + \frac{C}{B}$$

In this case in which $(4B - A^2) > 0$, we get a solution which approaches C/B as long as $A > 0$, so that the exponential factor $\exp(-A^*t/2)$ drives the oscillating sine and cosine term to zero for $t \gg 1/A$.

Let y_{pos} be an expression depending on t which is a time dependent solution depending on the constant real values of A , B , and C .

(%i41) $y_{pos} : \text{rhs}(\text{gsoln});$

$$(\text{ypos}) \quad \%e^{-\frac{A t}{2}} \left(\%k1 \sin\left(\frac{\sqrt{4 B - A^2} t}{2}\right) + \%k2 \cos\left(\frac{\sqrt{4 B - A^2} t}{2}\right) \right) + \frac{C}{B}$$

Here we make sure (OCD syndrome) this solution satisfies the given ODE.

(%i42) $\text{check}(y_{pos}, t);$

(%o42) 0

An alternative approach (in Maxima) is to design a Maxima function which requires us to decide (ahead of time) on the relative sizes of A, B, C (or express them symbolically) and ask for the specific cases: complex roots (sines and cosines) if $A^2 < 4B$, distinct real roots (if $A^2 > 4B$), or double real roots (if $A^2 = 4B$). For the arg 'type' in Lode2 then use either complex, real, or double.

We showed an example of using Lode2 already in the "Fast Track" section at the top. You need to load Econ2.mac first (which we did at the top of this worksheet).

(%i43) $\text{Lsoln} : \text{Lode2}(y, t, \text{complex}, A, B, C);$

this assumes that $4 B > A^2$

$$(\text{Lsoln}) \quad y = \%e^{-\frac{A t}{2}} \left(\%k1 \sin\left(\frac{\sqrt{4 B - A^2} t}{2}\right) + \%k2 \cos\left(\frac{\sqrt{4 B - A^2} t}{2}\right) \right) + \frac{C}{B}$$

5.1.1 Example: $y'' + A y' + y = 0, A^2 < 4$

We assume here that $C = 0, B = 1$, and $\%k1 = \%k2 = 1$.

Since $C = 0, y_p = 0$. Also, since we are assuming (above) that $4B > A^2$, the roots are complex and we get sines and cosines modulated by an exponential factor. $B = 1$, together with $4B > A^2$, implies that $0 < A < 2$ or $-2 < A < 0$.

We first use our Maxima function Lode2 and specialize to $C = 0, B = 1$.

```
(%i44) Lode2(y,t,complex,A,1,0);
```

this assumes that $4 > A^2$

```
(%o44) y=%e- $\frac{A t}{2}$ 

$$\left( \%k1 \sin\left(\frac{\sqrt{-(A-2)(A+2)} t}{2}\right) + \%k2 \cos\left(\frac{\sqrt{-(A-2)(A+2)} t}{2}\right) \right)$$

```

A global flag dofactor is set to true in Econ2.mac

```
(%i45) dofactor;
```

```
(%o45) true
```

If we change that setting of dofactor here, we get an unfactored argument of the square root function sqrt.

```
(%i47) dofactor : false$
```

```
Lode2(y,t,complex,A,1,0);
```

this assumes that $4 > A^2$

```
(%o47) y=%e- $\frac{A t}{2}$ 

$$\left( \%k1 \sin\left(\frac{\sqrt{4-A^2} t}{2}\right) + \%k2 \cos\left(\frac{\sqrt{4-A^2} t}{2}\right) \right)$$

```

```
(%i48) grind(%)$
```

```
y = %e-((A*t)/2)*(%k1*sin((sqrt(4-A^2)*t)/2)+%k2*cos((sqrt(4-A^2)*t)/2))$
```

Next we use Maxima's solution from ode2 (gsoln and ypos above) and specialize to the case $C = 0$, $B = 1$, $\%k1 = \%k2 = 1$. The name "ypos" reflects our assumption that $4*B > A^2$ in this section, corresponding to the question ode2 asks.

We can use the at (...) function to define yex for the case $C = 0$, $B = 1$, $\%k1 = \%k2 = 1$.

```
(%i49) yex : at (ypos, [C = 0, B = 1, %k1 = 1, %k2 = 1]);
```

```
(yex) %e- $\frac{A t}{2}$ 

$$\left( \sin\left(\frac{\sqrt{4-A^2} t}{2}\right) + \cos\left(\frac{\sqrt{4-A^2} t}{2}\right) \right)$$

```

In economics, we are interested (usually) in positive times (the future). What is the value of yex at time $t = 0$? Again we can use the at function.

```
(%i50) at(yex, t = 0);
```

```
(%o50) 1
```

So if we vary the parameter A, this solution always starts at $y(0) = 1$.

We now define a Maxima function depending on (t, A) using TWO single quotes (') in front of yex (to force evaluation of the expression yex):

```
(%i51) ys (t, A) := "yex;
```

```
(%o51) ys(t,A) := %e-A t / 2  $\left( \sin\left(\frac{\sqrt{4-A^2} t}{2}\right) + \cos\left(\frac{\sqrt{4-A^2} t}{2}\right) \right)$ 
```

The $t = 0$ value of $y(t)$, for any real value of A (satisfying $A^2 < 4$) can then be found using this function $ys(t, A)$.

```
(%i52) ys(0,A);
```


```
(%o52) 1
```

The Maxima function ys can now also be used to show plots for different values of A.

Positive values of A, respecting $(4 - A^2) > 0$, or $A^2 < 4$ here, show $ys(t)$ converging to $C/B = 0$, as seen in this plot. The overall factor $\exp(-A t/2)$, which starts out with unit value at $t = 0$, approaches zero rapidly for "large" positive values of t, ($t \gg 1/A$).

```
(%i53) wxdraw2d (title = "y'' + A y' + y = 0, 0 < A < 2", xlabel = "t", ylabel = "y",
  key = "A = 0.5", explicit (ys(t, 0.5), t, 0, 20),
  key = "A = 0.25", color = brown, explicit (ys(t, 0.25), t, 0, 20),
  key = "A = 0.125", color = red, explicit (ys(t, 0.125), t, 0, 20),
  color = black, key = "", line_width = 1, explicit (0, t, 0, 20))$
```

```
(%t53)
```




Larger positive values of A lead to faster convergence to the equilibrium value $y = 0$.

Next we look at negative values of A. In this case the overall factor $\exp(-A^*t/2)$ increases explosively away from unit value (at $t = 0$) for positive times t . Negative values of A, respecting $(4*B - A^2) > 0$, or $A^2 < 4$ here, show $y_s(t)$ diverging (in absolute value) away from $y(0) = 1$.

```
(%i55) tmax : 15;
wxdraw2d (title = " y'' + A y' + y = 0, -2 < A < 0",
  xlabel = "t", ylabel = "y", yrange = [-20,20], key_pos = bottom_left,
  key = "A = -0.5", explicit (ys(t, -0.5), t, 0, tmax),
  key = "A = -0.25", color = brown, explicit (ys(t, -0.25), t, 0, tmax),
  key = "A = -0.125", color = red, explicit (ys(t, -0.125), t, 0, tmax),
  color = dark_cyan, key = "y(0) = 1", explicit (1, t, 0, 0.75),
  color = black, key = "", line_width = 1, explicit (0, t, 0, tmax))$
```

(tmax) 15



Each case of negative A shows divergence from $y = 0$. The greater the magnitude of A, the faster the divergence.

5.1.2 Example: $y'' + y' + B y = 0, B > 1/4$

A second example of a linear second order ODE with constant coefficients and term, which has complex characteristic roots is $y'' + y' + B y = 0$, as long as $B > 1/4$.

```
(%i56) Lode2 (y,t,complex,1,B,0);
```

this assumes that $4 B > 1$

```
(%o56) y=%e-t/2 ( %k1 sin( (sqrt(4 B-1) t) / 2 ) + %k2 cos( (sqrt(4 B-1) t) / 2 ) )
```

Again, for practice using results from the chain starting with ode2, we redefine yex by specializing ypos (defined above) for this case, including unit values for %k1 and %k2.

```
(%i57) yex : at (ypos, [C = 0, A = 1, %k1 = 1, %k2 = 1]);
```

```
(yex) %e- $\frac{t}{2}$   $\left( \sin\left(\frac{\sqrt{4B-1}t}{2}\right) + \cos\left(\frac{\sqrt{4B-1}t}{2}\right) \right)$ 
```

```
(%i58) yex;
```

```
(%o58) %e- $\frac{t}{2}$   $\left( \sin\left(\frac{\sqrt{4B-1}t}{2}\right) + \cos\left(\frac{\sqrt{4B-1}t}{2}\right) \right)$ 
```

This is clearly damped for positive $t \gg 1$.

We again use two single quote marks (') in front of the Maxima expression yex to define a Maxima function ysf (t, B).

```
(%i59) ysf (t,B) := "yex;
```

```
(%o59) ysf (t,B) := %e- $\frac{t}{2}$   $\left( \sin\left(\frac{\sqrt{4B-1}t}{2}\right) + \cos\left(\frac{\sqrt{4B-1}t}{2}\right) \right)$ 
```

```
(%i60) ysf(t, B);
```

```
(%o60) %e- $\frac{t}{2}$   $\left( \sin\left(\frac{\sqrt{4B-1}t}{2}\right) + \cos\left(\frac{\sqrt{4B-1}t}{2}\right) \right)$ 
```


The initial value ($t = 0$) for any $B > 1/4$ is found from this function using ys(0,B):

```
(%i61) ysf(0,B);
```

```
(%o61) 1
```

```
(%i63) tmax : 10$
wxdraw2d ( xlabel = "t", ylabel = "y",yrange = [-1, 1.5],
title = " y'' + y' + B*y = 0, B > 1/4",
key = "B = 0.3", explicit (ysf(t, 0.3), t, 0, tmax),
key = "B = 0.5", color = red, explicit (ysf(t, 0.5), t, 0, tmax),
key = "B = 1", color = dark_cyan, explicit (ysf(t, 1), t, 0, tmax),
key = "B = 5", color = brown, explicit (ysf(t, 5), t, 0, tmax),
color = black,key = "", line_width = 1, explicit (0,t,0,tmax))$
```

(%t63)



Large positive values of B produce clear damped oscillations, small values of B > 1/4 are also damped, but without much oscillation.

5.2 Case: A^2 > 4*B, Distinct Real Roots

This is the more complicated case, in which the characteristic roots r1 and r2 are real. Here (A^2 - 4*B) > 0 and the solution is expressed in terms of sqrt(A^2 - 4*B).

```
(%i64) Lode2 (y, t, real, A, B, C);
```

this assumes that A^2 > 4 B

$$y = k_2 e^{-\frac{(\sqrt{A^2 - 4B} + A)t}{2}} + k_1 e^{\frac{(\sqrt{A^2 - 4B} - A)t}{2}} + \frac{C}{B}$$

```
(%i65) gsoln : ode2 (eqn,y,t);
```

Is 4 B - A^2 positive, negative or zero? negative;

$$y = k_1 e^{\frac{(\sqrt{A^2 - 4B} - A)t}{2}} + k_2 e^{\frac{(-\sqrt{A^2 - 4B} - A)t}{2}} + \frac{C}{B}$$


```
(%i66) yneg : rhs(gsoln);
```

```
(%neg) %k1 %e(sqrt(A^2-4 B)-A) t / 2 + %k2 %e(-sqrt(A^2-4 B)-A) t / 2 + C/B
```

```
(%i67) check (yneg, t);
```

```
(%o67) 0
```

5.2.1 Example: $y'' + A y' + y = 0$, $A^2 > 4$

```
(%i68) Lode2 (y,t,real,A,1,0);
```

this assumes that $A^2 > 4$

```
(%o68) y=%k2 %e(sqrt(A^2-4)+A) t / 2 + %k1 %e(sqrt(A^2-4)-A) t / 2
```

$C = 0$ implies that $y_p = 0$, and $y = y_c$.

$B = 1$, together with $A^2 > 4B$, implies that $A > 2$ or $A < -2$.

We also assume $\%k1 = \%k2 = 1$.

```
(%i69) yex : at (yneg, [C = 0, B = 1, %k1 = 1, %k2 = 1]);
```

```
(%yex) %e(sqrt(A^2-4)-A) t / 2 + %e(-sqrt(A^2-4)-A) t / 2
```

```
(%i70) at (yex, t = 0);
```

```
(%o70) 2
```

Define a Maxima function $ysf(t,A)$ which can be used for A either positive or negative, as long as $A^2 > 4$.

```
(%i72) kill(ysf)$
```

```
ysf (t, A) := "yex;
```

```
(%o72) ysf (t, A) := %e(sqrt(A^2-4)-A) t / 2 + %e(-sqrt(A^2-4)-A) t / 2
```

We first consider the case that $A > 2$, so A is positive.

Since $\sqrt{A^2 - 4} < A$, the first exponential term has the form $\exp(-r^*t)$ with $r > 0$, $t > 0$, which rapidly becomes vanishingly small as t gets "large". The second exponential term also has the form $\exp(-r^*t)$ with both r and t positive and goes to zero even faster than the first term. For longer times the first term dominates and is zero for $t \gg 1/(A - \sqrt{A^2 - 4})$,


Initial value of y at $t = 0$, for any $A > 2$:

```
(%i73) ysf(0,A);
```

```
(%o73) 2
```

```
(%i74) wxdraw2d (xlabel = "t", ylabel = "y", yrange = [0.2, 2],
  title = "y'' + A y' + y = 0, A > 2",
  key = "A = 3", explicit (ysf (t, 3), t, 0, 3),
  color = brown, key = "A = 6", explicit (ysf (t, 6), t, 0, 3),
  color = red, key = "A = 12", explicit (ysf (t, 12), t, 0, 3))$
```

```
(%t74)
```



We see convergence to $C/B = 0$ as long as $A > 2$, with faster convergence for values of A closer to 2. $ysf(t)$ gets close to C/B for $t \gg 1/(A - \sqrt{A^2 - 4})$.

We next consider negative values of A , respecting $A < -2$ (so that $A^2 > 4$).


The first term of $ysf(t, A)$ is $\exp(r^*t)$ with $r = (\sqrt{A^2 - 4} - A) = (\sqrt{A^2 - 4} + |A|) > 0$.

The second term of $ysf(t, A)$ is $\exp(r^*t)$ with $r = -\sqrt{A^2 - 4} - A = (|A| - \sqrt{A^2 - 4}) > 0$.

Both terms grow explosively.

```
(%i76) tmax : 0.5;
wxdraw2d (xlabel = "t", ylabel = "y", yrange = [0, 20],
title = "y'' + A y' + y = 0, A < - 2 ",
key_pos = top_left,
key = "A = -3", explicit (ysf(t, -3), t, 0, tmax),
color = brown, key = "A = -6", explicit (ysf(t, -6), t, 0, tmax),
color = red, key = "A = -9", explicit (ysf(t, -9), t, 0, tmax),
color = dark_cyan, key = "y(0) = 2", explicit (2, t, 0, tmax))$
```

(tmax) 0.5



(%t76)

We see exponential growth in $y(t)$ away from $y(0) = 2$, with faster growth occurring with more negative values of $A < -2$ (greater in magnitude).

5.2.2 Example: $y'' + y' + B y = 0, B < 1/4$

This is a second example of distinct real roots.

$C = 0$ implies that $y_p = 0$, and $y = y_c$.

$A = 1$, together with $A^2 > 4 \cdot B$, implies that $1 > 4 \cdot B$, or $B < 1/4$.

We also assume $k_1 = k_2 = 1$.

```
(%i77) Lode2 (y, t, real, 1, B, 0);
```

this assumes that $1 > 4 B$

```
(%o77) y = %k2 %e $\frac{(\sqrt{1-4B}+1)t}{2}$  + %k1 %e $\frac{(\sqrt{1-4B}-1)t}{2}$ 
```

```
(%i78) yex : at (yneg, [C = 0, A = 1, %k1 = 1, %k2 = 1]);
```

```
(yex) %e  $\frac{(\sqrt{1-4B}-1)t}{2}$  + %e  $\frac{(-\sqrt{1-4B}-1)t}{2}$ 
```

```
(%i79) at (yex, t = 0);
```

```
(%o79) 2
```

```
(%i81) kill(ysf)$
```

```
ysf (t, B) := "yex;
```

```
(%o81) ysf (t, B) := %e  $\frac{(\sqrt{1-4B}-1)t}{2}$  + %e  $\frac{(-\sqrt{1-4B}-1)t}{2}$ 
```


```
(%i82) ys(0,B);
```

```
(%o82) 1
```

```
(%i84) tmax : 10$
```

```
wxdraw2d (xlabel = "t", ylabel = "y", xrange = [0, 2.1],
  key_pos = bottom_left,
  title = "y'' + y' + B y = 0, B < 1/4",
  key = "B = 0.1", explicit (ysf (t, 0.1), t, 0, tmax),
  color = dark_cyan, key = "B = -0.1", explicit (ysf (t, -0.1), t, 0, tmax),
  color = brown, key = "B = -0.2", explicit (ysf (t, -0.2), t, 0, tmax),
  color = red, key = "B = -0.3", explicit (ysf (t, -0.3), t, 0, tmax))$
```

```
(%t84)
```



The case $B = 0.1$ illustrates the result that if $0 < B < 1/4$, $ysf(t)$ has no minimum value. The other three cases have B negative, and we see that $ysf(t)$ has a minimum value for some t .

5.2.3 Case: $4B = A^2$, Double Real Root, Examples $A > 0$, $A < 0$

When $4B = A^2$, the solution has both linear and exponential factors, which sometimes leads to a curve with a point of inflection. With $B = A^2/4$, the characteristic equation becomes

$$r^2 + Ar + A^2/4 = 0, \text{ or } (r + A/2)^2 = 0, \text{ or } r = -A/2.$$

(%i85) factor ($r^2 + A*r + A^2/4$);

(%o85)
$$\frac{(2r+A)^2}{4}$$

(%i86) solve ($r^2 + A*r + A^2/4, r$);

(%o86)
$$\left[r = -\frac{A}{2} \right]$$

(%i87) Lode2(y,t,double,A,A^2/4,C);

this assumes that $A^2 = A^2$

(%o87)
$$y = (\%k2 t + \%k1) e^{-\frac{A t}{2}} + \frac{4 C}{A^2}$$

(%i88) gsoln : ode2 (eqn, y, t);

Is $4B - A^2$ positive, negative or zero? nonzero;

Is $4B - A^2$ positive, negative or zero? zero;

Is A zero or nonzero? nonzero;

(gsoln)
$$y = (\%k2 t + \%k1) e^{-\frac{A t}{2}} + \frac{4 C}{A^2}$$

(%i89) yzero : rhs(gsoln);

(yzero)
$$(\%k2 t + \%k1) e^{-\frac{A t}{2}} + \frac{4 C}{A^2}$$

(%i90) yex : at (yzero, [C = 0, %k1 = 1, %k2 = 1]);


(yex)
$$(t+1) e^{-\frac{A t}{2}}$$

(%i92) kill(ysf)\$

ysf (t, A) := "yex;

(%o92)
$$ysf(t, A) := (t+1) e^{-\frac{A t}{2}}$$

```
(%i94) tmax : 10$
wxdraw2d (xlabel = "t", ylabel = "y", yrange = [0, 1.4],
title = "y'' + A y' + (A^2/4) y = 0, A > 0 ",
key = "A = 1", explicit (ysf (t, 1), t, 0, tmax),
color = brown, key = "A = 2", explicit (ysf (t, 2), t, 0, tmax),
color = red, key = "A = 4", explicit (ysf (t, 4), t, 0, tmax))$
```



(%t94)

A numerical example in which $A = -4$, and we vary the relative sizes of the constants k_1, k_2 :

```
(%i95) yex : at (yzero, [C = 0, A = -4]);
```


```
(yex) (%k2 t + %k1) %e2 t
```

```
(%i97) kill(ysf)$
ysf(t, %k1, %k2) := "yex;
```

```
(%o97) ysf (t, %k1, %k2) := (%k2 t + %k1) %e2 t
```

We now consider positive and negative combinations of k_1 and k_2 over the t interval $[0, 1]$.

```
(%i98) wxdraw2d (xlabel = "t", ylabel = "y", yrange = [-15, 15],
    title = " y'' - 4 y' + 4y = 0 ",
    key_pos = top_left,
    key = "k1=1, k2 = 1", explicit (ysf (t,1, 1), t, 0, 1),
    color = brown, key = "k1=1, k2 = -1", explicit (ysf (t,1, -1), t, 0, 1),
    color = red, key = "k1= -1, k2 = 1", explicit (ysf (t, -1, 1), t, 0, 1),
    color = magenta, key = "k1= -1, k2 = -1", explicit (ysf (t, -1, -1), t, 0, 1),
    color = black, key = "", line_width = 1, explicit (0, t, 0, 1))$
```



(%t98)

5.2.4 Case: B = 0, Dowling 18.4

Consider the equation

$$y'' + 7 y' = 28$$

for which A = 7, B = 0, C = 28, A^2 = 49 > 4*B = 0.

```
(%i100) eqn : 'diff(y,t,2) + 7*'diff(y,t) -28;
    gsoln : ode2(eqn,y,t);
```

(eqn) $\frac{d^2}{d t^2} y + 7 \left(\frac{d}{d t} y \right) - 28$

(gsoln) $y = \%k2 \%e^{-7 t} + \frac{28 t - 4}{7} + \%k1$

If A or B is zero, Lode2 automatically calls ode2.

```
(%i101) Lode2(y,t,real,7,0,28);
```

(%o101) $y = \%k2 \%e^{-7 t} + \frac{28 t - 4}{7} + \%k1$

(%i102) expand(%);

(%o102) $y = k_2 e^{-7t} + 4t + k_1 - \frac{4}{7}$

Naturally, one can combine k_1 and the term $-4/7$ into some other adjustable constant, say A_1 , and write this as $y = A_1 + 4t + A_2 \exp(-7t)$, as in Dowling.

6 Market Model with Price Expectations

We adapt Sec. 16.4 in Chiang and Wainwright.

In prior chapters the quantity demanded Q_d and the quantity Q_s supplied were assumed to depend only on the current price P . We can relax that assumption by recognising that both buyers and sellers often are influenced by the "price trend" prevailing at a given time t . The current price trend can induce expectations about the price level in the future, and affect either (or both) Q_d and Q_s .

In the continuous-time context, price trend information comes mainly from the current value of both the current rate of change of the price, $P' = dP/dt$, and the current rate of change of dP/dt , $P'' = d^2P(t)/dt^2$.

A continuous time market model including the influence of the current price trend (in this context) would require a model of the form:

$$\begin{aligned} Q_d &= D[P(t), P'(t), P''(t)], \\ Q_s &= S[P(t), P'(t), P''(t)]. \end{aligned}$$

If we specialize to the linear version of these functions D and S , expressed in terms of constant coefficients b, m, n, d, u, w and terms a, g :

$$\begin{aligned} Q_d &= a - bP + mP' + nP'', \text{ with } a, b > 0, \\ Q_s &= -g + dP + uP' + wP'', \text{ with } g, d > 0. \end{aligned}$$

The price trend parameters, whose signs have not been restricted, depend on the future price expectations of the buyers (m and n) and the sellers (u and w).

If $m > 0$, a rising price ($P' > 0$) will result in an increase of Q_d - buyers expect the price to continue to rise significantly and opt to make purchases at the current time t , while the price is affordable. The magnitude of m , in both sign cases, would reflect the strength of the corresponding expectations.

If $m < 0$, a rising price ($P' > 0$) will result in a decrease of Q_d - buyers expect a quick reversal of the price rise behavior and decide to stop current purchases and wait for a lower price level to appear.

The inclusion of the parameter n makes the buyers' behavior depend also on the rate of change of dP/dt , and both parameters m and n include the effects of price speculation on the behavior of the market price in this model.

The signs and magnitudes of the supply side parameters u and w likewise would depend on sellers' expectations of future price levels.

6.1 A Simplified Model

Assume here that we only need to include price trend assumptions in the demand function expression Q_d :

$$Q_d = a - bP + mP' + nP'', \text{ with } a, b > 0,$$

$$Q_s = -g + dP, \text{ with } g, d > 0$$

Assume the market is cleared at every point of time (continuous time model):

$$Q_d = Q_s, \text{ or } Q_d - Q_s = 0.$$

As a help for algebraic manipulation let P_1 stand for P' and P_2 stand for P'' .

```
(%i106) kill(a,b,m,n,g,d)$
Qd : a - b*P + m*P1 + n*P2;
Qs : - g + d*P;
aeqn : Qd - Qs;
(Qd)  P2 n+P1 m-P b+a
(Qs)  P d-g
(aeqn) P2 n+P1 m+g-P d-P b+a
```

We want to arrive at the 2nd order ODE form

$$P'' + A P' + B = C, \text{ or } P'' + A P' + B - C = 0$$

which we adopted in prior sections. So divide by the parameter n .

```
(%i107) factorsum(aeqn);
(%o107) P2 n+P1 m+g-P (d+b) +a

(%i108) %/n;
(%o108) 
$$\frac{P_2 n + P_1 m + g - P (d + b) + a}{n}$$

```

Dividing by n we get the form

$$P'' + A P' + B P - C = 0$$

with

$$A = m/n, \quad B = -(b + d)/n, \quad \text{and} \quad C = -(a + g)/n.$$

When $P'' = 0$ and $P' = 0$, we reach "intertemporal equilibrium" $P_e = C/B$ (perhaps only for a moment, if this is an unstable equilibrium),

$$P_e = (a+g)/(b+d) > 0.$$

We get two distinct real roots if $A^2 > 4*B$, or if $(m/n)^2 > -4*(b+d)/n$.

If $n > 0$, $-4*(b+d)/n$ is a negative number (since b and d are positive), and we certainly get two distinct real roots. What about stability?

Without assuming a sign for n , Maxima's solve function gets the roots r_1 and r_2 in the form:

```
(%i110) assume(b > 0, d > 0)$
[r1, r2] : map ('rhs, solve (r^2 + (m/n)*r - (b+d)/n = 0,r) );
(%o110) [-\frac{\sqrt{(4 d+4 b) n+m^2}+m}{2 n}, \frac{\sqrt{(4 d+4 b) n+m^2}-m}{2 n}]
```

If we assume $n > 0$, the expression under the radical is both positive and greater than m^2 ; $[4*(b+d)*n + m^2] > m^2$, so its square root is greater than the magnitude of m , so $r_1 < 0$ and $r_2 > 0$, no matter the sign of m . So $n > 0$ only leads to distinct real roots.

With one negative root and one positive root, there is no stable intertemporal equilibrium possible (except for the case that the initial conditions exclude the effect of the positive root).

If we assume $n < 0$, then all three cases become possible:

real roots if $m^2 > 4*(b+d)*|n|$, in which case $0 < \text{sqrt}(4*(b+d)*n + m^2) < |m|$

complex roots if $m^2 < 4*(b+d)*|n|$.

We get complex roots r_1 and r_2 if $A^2 < 4*B$, or if $(m/n)^2 < -4*(b+d)/n$, or (multiplying both sides by n^2) if $m^2 < -4*(b+d)*n$, or $(m^2 + 4*(b+d)*n) < 0$, which is not possible if n is positive; we need $n < 0$ to get complex roots.

6.1.1 Distinct Real Roots Case: $(m/n)^2 > -4*(b+d)/n$

Distinct real roots occur if $A^2 > 4*B$ (from our previous discussion), or if $(m/n)^2 > -4*(b+d)/n$, which is ALWAYS true if $n > 0$, and MAY be true if $n < 0$. In this case $P(t) = A1*\exp(r1*t) + A2*\exp(r2*t) + Pe$, with the signs and magnitudes of the real roots $r1$ and $r2$ dependent of the values of A and B .

The distinct real roots $r1, r2$ are the roots of the equation $r^2 + A*r + B = 0$.

```
(%i111) solve (r^2 + (m/n)*r - (b+d)/n = 0,r);
```

```
(%o111) [r = -\frac{\sqrt{(4 d+4 b) n+m^2}+m}{2 n}, r = \frac{\sqrt{(4 d+4 b) n+m^2}-m}{2 n}]
```

```
(%i112) [r1, r2] : map ('rhs, %);
```

```
(%o112) [-\frac{\sqrt{(4 d+4 b) n+m^2}+m}{2 n}, \frac{\sqrt{(4 d+4 b) n+m^2}-m}{2 n}]
```

```
(%i113) yreal : %k1*exp(r1*t) + %k2*exp(r2*t) + (a+g)/(b+d);
```

```
(yreal) %k1 %e^{-\frac{(\sqrt{(4 d+4 b) n+m^2}+m) t}{2 n}} + %k2 %e^{\frac{(\sqrt{(4 d+4 b) n+m^2}-m) t}{2 n}} + \frac{g+a}{d+b}
```

6.1.2 Example 1

Consider the numerical example

$$Q_d = 42 - 4P - 4P' + P''$$

$$Q_s = -6 + 8P$$

with the initial conditions $P(0) = 6, P'(0) = 4$.

```
(%i114) case1 : [a=42,b=4,m=-4,n=1,g=6,d=8]$
```

```
(%i115) at ([r1, r2], case1);
```

```
(%o115) [-2, 6]
```

```
(%i116) yex1 : at (yreal, case1);
```

```
(yex1) %k2 %e^{6 t} + %k1 %e^{-2 t} + 4
```

```
(%i117) ic2(P = yex1, t = 0, P = 6, 'diff(P,t) = 4);
```

```
(%o117) P = %e^{6 t} + %e^{-2 t} + 4
```

The existence of the positive root $r2 = 6$ means that the intertemporal equilibrium $Pe = 4$ is dynamically unstable.

```
(%i118) y1 : rhs(%);
```


```
(y1) %e6 t+%e-2 t+4
```

```
(%i119) at (y1, t = 0);
```

```
(%o119) 6
```

```
(%i120) wxdraw2d ( xlabel = "t", ylabel = "P", yrange = [0, 100],
  title = "P = exp(6*t) + exp(-2*t) + 4",
  explicit (y1,t,0,1),color = red, line_width = 1,
  key_pos = top_left, key = "P(0) = 6", explicit (6,t,0,0.2))$
```

```
(%t120)
```



6.1.3 Complex Roots Case: $(m/n)^2 < -4*(b+d)/n$

Assume the demand and supply functions are

$$Q_d = 40 - 2P - 2P' - P'' = a - bP + mP' + nP'', \quad m/n = (-2)/(-1) = 2, \quad A^2 = 4.$$

$$Q_s = -5 + 3P = -g + dP,$$

$b = 2, d = 3, b+d = 5, n = -1, B = -(b+d)/n = -5/(-1) = 5, A^2 < B \Rightarrow$ complex roots,
Easiest to use ode2 directly.

Equate Q_d and Q_s and divide by n to get

$$P'' + (m/n)P' - (b+d)P/n = -(a+g)/n$$

$$P'' + 2P' + 5P = 45.$$

```
(%i122) eqn : 'diff (P,t,2) + 2*'diff(P,t) + 5*P = 45;
```

```
gsoln : ode2(eqn, P, t);
```

```
(eqn)  $\frac{d^2}{dt^2} P + 2 \left( \frac{d}{dt} P \right) + 5P = 45$ 
```

```
(gsoln)  $P = e^{-t} (\%k1 \sin(2t) + \%k2 \cos(2t)) + 9$ 
```

For $t \gg 1$ (say 5 or 10), $P \sim 9$.

6.1.4 Example 2

Assume $P(0) = 12$, $P'(0) = 1$.

```
(%i123) dsoln : ic2 (gsoln, t = 0, P = 12, 'diff (P,t) = 1);
```

```
(dsoln) P=%e-t (2 sin(2 t)+3 cos(2 t)) +9
```


```
(%i124) y2 : rhs (dsoln);
```

```
(y2) %e-t (2 sin(2 t)+3 cos(2 t)) +9
```

```
(%i125) at (y2, t = 0);
```

```
(%o125) 12
```

```
(%i126) wxdraw2d (xlabel = "t", ylabel = "P", yrange = [8,12.5],
explicit (y2, t, 0, 5), line_width = 1, color = red, key = "Pe = 9",
explicit (9, t, 0, 5))$
```



6.2 Including Supply Side Price Trend Expectations

Assume

$$Q_d = 9 - P + P' + 3 P'',$$

$$Q_s = -1 + 4 P - P' + 5 P'',$$

Equating Q_d and Q_s and writing in form $P'' + A P' + B P = C$,

we get $P'' - P' + (5/2) P = 5$, $A = -1$, $B = 5/2$, $C = 5$, $P_e = C/B = 2$, $A^2 = 1$, $4*B = 10$,

$A^2 < 4*B \Rightarrow$ complex roots.

```
(%i128) eqn: 'diff(P,t,2) - 'diff (P,t) + (5/2)*P = 5;
        gsoln : ode2 (eqn, P, t);
```

```
(eqn) 
$$\frac{d^2}{dt^2} P - \frac{d}{dt} P + \frac{5P}{2} = 5$$

```

```
(gsoln) 
$$P = \%e^{t/2} \left( \%k1 \sin\left(\frac{3t}{2}\right) + \%k2 \cos\left(\frac{3t}{2}\right) \right) + 2$$

```

We see a divergent price $P(t)$ predicted by this model.
The intertemporal equilibrium $P_e = 2$ is dynamically unstable.
Assume $P(0) = 4$ and $P'(0) = 4$.

```
(%i129) dsoln : ic2(gsoln, t = 0, P = 4, 'diff (P,t) = 4);
```

```
(dsoln) 
$$P = \%e^{t/2} \left( 2 \sin\left(\frac{3t}{2}\right) + 2 \cos\left(\frac{3t}{2}\right) \right) + 2$$

```

```
(%i130) y3 : rhs (dsoln);
```

```
(y3) 
$$\%e^{t/2} \left( 2 \sin\left(\frac{3t}{2}\right) + 2 \cos\left(\frac{3t}{2}\right) \right) + 2$$


```

```
(%i131) at (y3, t = 0);
```

```
(%o131) 4
```

```
(%i132) wxdraw2d (xlabel = "t", ylabel = "P", explicit (y3, t, 0, 2),
        color = black, line_width = 1, explicit (0, t, 0, 2))$
```

```
(%t132)
```



```
(%i133) solve (diff (y3,t) = 0, t);
```

```
(%o133) 
$$\left[ \sin\left(\frac{3t}{2}\right) = 2 \cos\left(\frac{3t}{2}\right) \right]$$

```

An extremum of $P(t)$ thus occurs at $\tan(3t/2) = 2$, or $t = (2/3)\text{arcTan}(2)$,


```
(%i134) tmax : 2.0*atan(2.0)/3;
```

```
(tmax) 0.7381
```

```
(%i135) Pmax : at (y3, t = tmax);
```

```
(Pmax) 5.881
```

```
(%i136) wxdraw2d (xlabel = "t", ylabel = "P", yrange = [-2, 6],
  explicit (y3, t, 0, 2), color = black, line_width = 1,
  explicit (0, t, 0, 2), color = red, key = "tmax = 0.738",
  parametric (tmax, yy, yy, -2, Pmax))$
```



7 Dowling Prob 18.36

In many markets supply and demand are influenced by current prices and price trends (i.e., whether prices are rising or falling and whether they are rising or falling at an increasing or decreasing rate). The economist, therefore, needs to know the current price $P(t)$, the first derivative $dP(t)/dt$, and the second derivative $d^2P(t)/dt^2$.

Assume the quantity supplied $Q_s(t)$ at time t depends on price $P(t)$ subject to

$$Q_s = c_1 + w_1 P(t) + u_1 P'(t) + v_1 P''(t) \text{ with } c_1, w_1, u_1, v_1 \text{ constants,}$$

and assume the quantity demanded $Q_d(t)$ at time t depends on price subject to

$$Q_d = c_2 + w_2 P(t) + u_2 P'(t) + v_2 P''(t) \text{ with } c_2, w_2, u_2, v_2 \text{ constants.}$$

Assuming price clears the market at each time t , $Q_s(t) = Q_d(t)$, and we get a second order ODE for $P(t)$. Let $c = c_1 - c_2$, $w = w_1 - w_2$, $u = u_1 - u_2$, $v = v_1 - v_2$. Then we get, after dividing by $v \neq 0$:

$$P''(t) + (u/v) P'(t) + (w/v) P(t) = -c/v.$$

Since $c = c_1 - c_2$ and $w = w_1 - w_2$, where under ordinary supply conditions $c_1 < 0$ & $w_1 > 0$, and under ordinary demand conditions $c_2 > 0$ and $w_2 < 0$, c is negative and w is positive, so $-c/w$ is positive and is the predicted possible equilibrium price P_e found by setting $P''(t) = 0$, $P'(t) = 0$, leaving $(w/v) P(t) = -c/v$, or $P(t) = -c/w = \text{positive constant} = P_e$.

Here we ask ode2 to find the general solution. In answer to the question: Is $4vw - u^2$ positive, negative or zero, we first answer positive, or simply p.

```
(%i139) kill(P,u,v,w,c)$
eqn : 'diff (P,t,2) + (u/v)*diff (P,t) + (w/v)*P = -c/v;
gsoln : ode2 (eqn, P, t);
```

$$(eqn) \quad \frac{P w}{v} + \frac{\left(\frac{d}{dt} P\right) u}{v} + \frac{d^2}{dt^2} P = -\frac{c}{v}$$

Is $4vw - u^2$ positive, negative or zero? positive;

$$(gsoln) \quad P = \%e^{-\frac{t u}{2 v}} \left(\%k1 \sin \left(\frac{t \sqrt{\frac{4 w}{v} - \frac{u^2}{v^2}}}{2} \right) + \%k2 \cos \left(\frac{t \sqrt{\frac{4 w}{v} - \frac{u^2}{v^2}}}{2} \right) \right) - \frac{c}{w}$$

Likewise, you can get other solutions by answering negative or zero.

```
(%i140) gsoln : ode2 (eqn, P, t);
```

Is $4vw - u^2$ positive, negative or zero? negative;

$$(gsoln) \quad P = \%k1 \%e^{\frac{t \left(\sqrt{\frac{u^2}{v^2} - \frac{4 w}{v}} - \frac{u}{v} \right)}{2}} + \%k2 \%e^{\frac{t \left(-\sqrt{\frac{u^2}{v^2} - \frac{4 w}{v}} - \frac{u}{v} \right)}{2}} - \frac{c}{w}$$

```
(%i141) gsoln : ode2 (eqn, P, t);
```

Is $4vw - u^2$ positive, negative or zero? zero;

Is u zero or nonzero? nonzero;

$$(gsoln) \quad P = (\%k2 t + \%k1) \%e^{-\frac{t u}{2 v}} - \frac{4 c v}{u^2}$$

8 Differential Equations with a Variable Term

We adapt the discussion in Chiang and Wainwright, Sec. 16.6. The following examples show the use of ode2 to find a general solution. These are all second order ODE's.

8.1 Example 1

General solution of
 $y'' + 5y' = 3y = 6t^2 - t - 1$.

```
(%i143) eqn : 'diff (y,t,2) + 5*'diff (y,t) + 3*y = 6*t^2 - t - 1;
        gsoln : ode2 (eqn, y, t);
```

$$(eqn) \quad \frac{d^2}{dt^2} y + 5 \left(\frac{d}{dt} y \right) + 3y = 6t^2 - t - 1$$

$$(gsoln) \quad y = \%k1 \%e^{\frac{(\sqrt{13}-5)t}{2}} + \%k2 \%e^{\frac{(-\sqrt{13}-5)t}{2}} + 2t^2 - 7t + 10$$

8.2 Example 2

General solution of $y'' + 5y' = 6t^2 - t - 1$.

```
(%i145) eqn : 'diff (y,t,2) + 5*'diff (y,t) = 6*t^2 - t - 1;
        gsoln : ode2 (eqn, y, t);
```

$$(eqn) \quad \frac{d^2}{dt^2} y + 5 \left(\frac{d}{dt} y \right) = 6t^2 - t - 1$$

$$(gsoln) \quad y = \%k2 \%e^{-5t} + \frac{500t^3 - 425t^2 - 80t + 16}{1250} + \%k1$$

8.3 Example 3

General solution of $y'' + 3y' - 4y = 2 \exp(-4t)$.

```
(%i147) eqn : 'diff (y,t,2) + 3*'diff (y,t) - 4*y = 2*exp (- 4*t);
        gsoln : ode2 (eqn, y, t);
```

$$(eqn) \quad \frac{d^2}{dt^2} y + 3 \left(\frac{d}{dt} y \right) - 4y = 2 \%e^{-4t}$$

$$(gsoln) \quad y = \%k1 \%e^t - \frac{(10t+2) \%e^{-4t}}{25} + \%k2 \%e^{-4t}$$

9 Higher Order ODE's With `desolve` (`eqn, [y(t)]`)

`ode2` only can only solve first or second order ODE's. If we try a 3rd order ODE, the `eqn` is returned and the return value is 'false'.

```
(%i149) eqn : 'diff (y,t,3) + 2*'diff (y,t,2) + 'diff (y,t) + 2*y = 8;
        gsoln : ode2 (eqn,y,t);
```

$$(eqn) \quad \frac{d^3}{dt^3} y + 2 \left(\frac{d^2}{dt^2} y \right) + \frac{d}{dt} y + 2 y = 8$$

$$(gsoln) \quad \frac{d^3}{dt^3} y + 2 \left(\frac{d^2}{dt^2} y \right) + \frac{d}{dt} y + 2 y = 8$$

not a proper differential equation

(gsoln) **false**

We can instead use `desolve`, which uses Laplace transform methods.

The syntax required in defining `eqn` is slightly different, since we must write the dependent variable as `y(t)` instead of just `y`. The following illustrates `desolve` syntax.

```
(%i151) eqn : 'diff (y(t),t,3) + 2*'diff (y(t),t,2) + 'diff (y(t),t) + 2*y(t) = 8;
        gsoln : desolve (eqn, [y(t)]);
```

$$(eqn) \quad \frac{d^3}{dt^3} y(t) + 2 \left(\frac{d^2}{dt^2} y(t) \right) + \frac{d}{dt} y(t) + 2 y(t) = 8$$

$$(gsoln) \quad y(t) = \frac{\sin(t) \left(2 \left(\frac{d^2}{dt^2} y(t) \Big|_{t=0} \right) + 5 \left(\frac{d}{dt} y(t) \Big|_{t=0} \right) + 2 y(0) - 8 \right)}{5} + \frac{\%e^{-2t} \left(\frac{d^2}{dt^2} y(t) \Big|_{t=0} + y(0) - 4 \right)}{5} - \frac{\cos(t) \left(\frac{d^2}{dt^2} y(t) \Big|_{t=0} - 4 y(0) + 16 \right)}{5} + 4$$

```
(%i152) grind(gsoln)$
```

$$y(t) = (\sin(t) * (2 * ('at('diff(y(t),t,2),t = 0)) + 5 * ('at('diff(y(t),t,1),t = 0)) + 2 * y(0) - 8))$$

/5

$$+ (\%e^{-(2*t)} * ('at('diff(y(t),t,2),t = 0) + y(0) - 4)) / 5$$

$$- (\cos(t) * ('at('diff(y(t),t,2),t = 0) - 4 * y(0) + 16)) / 5 + 4$$$

Use the `atvalue` function to set initial conditions and get a definite solution.

```
(%i156) atvalue ('diff (y(t),t,2), t = 0, 1);  
        atvalue ('diff (y(t), t), t = 0, 2);  
        atvalue (y(t), t = 0, 4);  
        dsoln : desolve (eqn, [y(t)]);
```

```
(%o153) 1
```

```
(%o154) 2
```

```
(%o155) 4
```

```
(dsoln)  $y(t) = \frac{12 \sin(t)}{5} - \frac{\cos(t)}{5} + \frac{e^{-2t}}{5} + 4$ 
```